

Aula prática 7

Comandos de repetição — for

Resumo

Nesta aula você desenvolverá algumas aplicações para treinar o uso do comando `for`.

Sumário

Tarefa 1: Valor de uma série

Desenvolva um programa Scilab para calcular o valor aproximado da série

$$\frac{1}{2} + \frac{1}{4} + \frac{1}{6} + \frac{1}{8} + \dots + \frac{1}{2*i} + \dots$$

O programa deverá solicitar ao usuário quantos termos devem ser usados no cálculo, e a seguir deve calcular e exibir o valor da série.

Exemplo de execução da aplicação

Cálculo da série

=====

Digite a quantidade de termos: 4

Valor da série com 4 parcelas: 1.04167

Solução:

```
clear;
clc;
printf("Cálculo da série\n");
printf("=====\n");
quant = input("Digite a quantidade de termos: ");
soma = 0;
for i = 1 : quant do
    soma = soma + 1/(2* i)
end
printf("Valor da série com %g parcelas: %g\n", quant, soma);
```

Tarefa 2: Valor de uma série

Faça um programa para calcular o valor de S , dado por

$$S = \frac{1}{n} + \frac{2}{n-1} + \frac{3}{n-2} + \dots + \frac{n-1}{2} + \frac{n}{1}$$

sendo n fornecido pelo usuário através do teclado.

Exemplo de execução da aplicação

Cálculo da série:

Digite a quantidade de parcelas: 100

Valor da série com 100 termos: 423.925

Solução:

```
clc;
clear;
printf("Cálculo da série:\n");
printf("-----\n");
n = input("Digite a quantidade de parcelas: ");
soma = 0;
for i = 1 : n
    soma = soma + i / (n - i + 1);
end
printf("Valor da série com %g termos: %g", n, soma);
```

Tarefa 3: Valor aproximado de π

Uma aproximação para o valor de π pode ser calculada pela seguinte fórmula:

$$\pi = \sqrt{12} \left(1 - \frac{1}{3 \times 3} + \frac{1}{5 \times 3^2} - \frac{1}{7 \times 3^3} + \frac{1}{9 \times 3^4} - \dots \right)$$

Quanto maior for o número de termos usados na soma da série, mais próximo do valor de π é o valor calculado. Isso é ilustrado na tabela exibida no exemplo de execução do programa, onde a primeira coluna mostra o número de termos usados no cálculo, a segunda coluna mostra o valor obtido usando esse número de termos e a terceira coluna mostra o erro absoluto entre o valor real (valor predefinido do Scilab) e valor calculado, isto é, módulo da diferença desses valores.

Escreva um programa que leia um valor inteiro positivo n , representando o número de parcelas a serem incluídas no somatório e imprima a tabela de execução do programa ilustrada abaixo.

Observação: Não utilizar vetores na implementação.

Exemplo de execução da aplicação

Cálculo aproximado de pi

Valor de pi pré-definido no scilab: 3.1415926536

Digite o número de termos da série: 10

termos	pi calculado	erro absoluto
--------	--------------	---------------

1	3.4641016151	0.32250896155
2	3.0792014357	0.06239121791
3	3.1561814716	0.01458881798
4	3.1378528916	0.00373976199
5	3.1426047457	0.00101209207
6	3.1413087855	0.00028386813
7	3.1416743127	0.00008165911
8	3.1415687159	0.00002393765
9	3.1415997738	0.00000712022
10	3.1415905109	0.00000214265

Dicas: Para calcular cada termo da série:

- use uma variável para representar o sinal do termo. O valor inicial desta variável deverá ser 1, e ela deverá ser invertida (multiplicada por -1) a cada passo do laço de repetição para que alterne entre 1 e -1.
- use uma variável para representar o primeiro fator do denominador. O valor inicial desta variável deverá ser 1, e ela deverá ser incrementada em duas unidades a cada passo do laço de repetição.
- use uma variável para representar expoente do segundo fator do denominador. O valor inicial desta variável deverá ser 0, e ela deverá ser incrementada em uma unidade a cada passo do laço de repetição.

Solução:

```
clc; clear;
printf("Cálculo aproximado de pi\n");
printf("-----\n");
printf("Valor de pi pré-definido no scilab: %.10f\n", %pi);
n = input("Digite o número de termos da série: ");
soma = 0;
sinal = 1;
x = 1;
y = 0;
printf("-----\n");
printf("%6s %13s %14s\n", "termos", "pi calculado", "erro absoluto");
printf("-----\n");
for i = 1:n
    soma = soma + sinal * 1 / (x * 3^y);
    piaprox = sqrt(12) * soma;
    erro = abs(piaprox - %pi);
    printf("%6i %13.10f %14.11f\n", i, piaprox, erro);
    sinal = - sinal;
    x = x + 2;
    y = y + 1;
end
```

Tarefa 4: Função de duas variáveis

Seja a função de duas variáveis:

$$f(x, y) = \begin{cases} x^2 - 3x + y^2 & \text{se } x < y \\ \frac{\sqrt{y^2 - 4x}}{2} & \text{se } x = y \\ \sqrt[3]{xy} & \text{se } x < y \end{cases}$$

Escreva um programa para gerar a tabela de valores dessa função (conforme o exemplo a seguir), para valores de x e y nos seguintes intervalos:

- $0 \leq x \leq 1$ (com incrementos de 0,1 em x)
- $0 \leq y \leq 1,4$ (com incrementos de 0,2 em y)

Exemplo de execução da aplicação

Tabela da função de duas variáveis

x/y	0.0	0.2	0.4	0.6	0.8	1.0	1.2	1.4
0.0	0.00	0.04	0.16	0.36	0.64	1.00	1.44	1.96
0.1	0.00	-0.25	-0.13	0.07	0.35	0.71	1.15	1.67
0.2	0.00	0.00	-0.40	-0.20	0.08	0.44	0.88	1.40
0.3	0.00	0.39	-0.65	-0.45	-0.17	0.19	0.63	1.15
0.4	0.00	0.43	0.00	-0.68	-0.40	-0.04	0.40	0.92
0.5	0.00	0.46	0.58	-0.89	-0.61	-0.25	0.19	0.71
0.6	0.00	0.49	0.62	0.00	-0.80	-0.44	0.00	0.52
0.7	0.00	0.52	0.65	0.75	-0.97	-0.61	-0.17	0.35
0.8	0.00	0.54	0.68	0.78	0.00	-0.76	-0.32	0.20
0.9	0.00	0.56	0.71	0.81	0.90	-0.89	-0.45	0.07
1.0	0.00	0.58	0.74	0.84	0.93	0.00	-0.56	-0.04

Observação: Não precisa exibir a primeira linha e a primeira coluna da tabela.

Solução:

```
clc; clear;
printf("Tabela da função de duas variáveis\n");
printf("-----\n");
printf("x/y |");
for y = 0:0.2:1.4
    printf(" %5.1f", y);
end
printf("\n");
printf("-----");
for y = 0:0.2:1.4
    printf("-----");
end
printf("\n");
for x = 0:0.1:1
    printf("%3.1f |", x);
    for y = 0:0.2:1.4
        if x < y then
            z = x^2 - 3*x + y^2;
        elseif x == y then
            z = sqrt(y^2 - 4*x)/2;
        else
            z = (x*y)^(1/3);
        end
        printf(" %5.2f", z);
    end
    printf("\n");
end
```